NAG Library Routine Document

F04AJF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F04AJF calculates the approximate solution of a set of real linear equations with multiple right-hand sides, $AX = B$, where A has been factorized by F03AFF.

2 Specification

```
SUBROUTINE F04AJF(N, IR, A, LDA, P, B, LDB)
INTEGER           N, IR, LDA, LDB
double precision  A(LDA,N), P(N), B(LDB,IR)
```

3 Description

To solve a set of real linear equations $AX = B$, F04AJF must be preceded by a call to F03AFF which computes an LU factorization of A with partial pivoting, $PA = LU$, where P is a permutation matrix, L is lower triangular and U is unit upper triangular. The columns x of the solution X are found by forward and backward substitution in $Ly = Pb$ and $Ux = y$, where b is a column of the right-hand sides.

4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer–Verlag

5 Parameters

1: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N $\geq$ 0.

2: IR – INTEGER *Input*

On entry: r, the number of right-hand sides.

3: A(LDA,N) – ***double precision*** array *Input*

On entry: details of the LU factorization, as returned by F03AFF.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F04AJF is called.

Constraint: LDA $\geq$ N.

5: P(N) – ***double precision*** array *Input*

On entry: details of the row interchanges as returned by F03AFF.

6: B(LDB,IR) – ***double precision*** array *Input/Output*

On entry: the n by r right-hand side matrix B.

On exit: B is overwritten by the solution matrix X.

7: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F04AJF is called.

Constraint: LDB $\geq$ N.

6 Error Indicators and Warnings

None.

7 Accuracy

The accuracy of the computed solutions depends on the conditioning of the original matrix. For a detailed error analysis see page 106 of Wilkinson and Reinsch (1971).

8 Further Comments

The time taken by F04AJF is approximately proportional to n^2r.

9 Example

This example solves the set of linear equations $AX = B$ where

$$A = \begin{pmatrix} 33 & 16 & 72 \\ -24 & -10 & -57 \\ -8 & -4 & -17 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} -359 \\ 281 \\ 85 \end{pmatrix}.$$

9.1 Program Text

```
*     F04AJF Example Program Text
*     Mark 14 Revised. NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NMAX, IR, LDA, LDB
      PARAMETER        (NMAX=8,IR=1,LDA=NMAX,LDB=NMAX)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      DOUBLE PRECISION D1, EPS
      INTEGER          I, ID, IFAIL, J, N
*     .. Local Arrays ..
      DOUBLE PRECISION A(LDA,NMAX), B(LDB,IR), P(NMAX)
*     .. External Functions ..
      DOUBLE PRECISION X02AJF
      EXTERNAL         X02AJF
*     .. External Subroutines ..
      EXTERNAL         F03AFF, F04AJF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F04AJF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      WRITE (NOUT,*)
      IF (N.GT.0 .AND. N.LE.NMAX) THEN
         READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
         IFAIL = 1
         EPS = X02AJF()
*
*        Crout decomposition
         CALL F03AFF(N,EPS,A,LDA,D1,ID,P,IFAIL)
*
         IF (IFAIL.EQ.0) THEN
            READ (NIN,*) ((B(I,J),J=1,IR),I=1,N)
```

```
*
*           Approximate solution of linear equations
            CALL F04AJF(N,IR,A,LDA,P,B,LDB)
*
            WRITE (NOUT,*) ' Solution'
            DO 20 I = 1, N
               WRITE (NOUT,99999) (B(I,J),J=1,IR)
   20       CONTINUE
         ELSE
            WRITE (NOUT,99998) IFAIL
         END IF
      ELSE
         WRITE (NOUT,99997) 'N is out of range: N = ', N
      END IF
*
99999 FORMAT (1X,8F9.4)
99998 FORMAT (1X,' ** F03AFF returned with IFAIL = ',I5)
99997 FORMAT (1X,A,I5)
      END
```

9.2 Program Data

```
F04AJF Example Program Data
  3
   33   16   72
  -24  -10  -57
   -8   -4  -17
 -359  281   85
```

9.3 Program Results

```
 F04AJF Example Program Results

  Solution
    1.0000
   -2.0000
   -5.0000
```